VNiVERSiDAD D SALAMANCA

Conferencia impartida en: Universidad Católica Santiago: Magister en Física Médica. Facultad de Física. 7 de junio de 2016

http://fisicamedica.uc.cl/index.php?option=com_content&view=article&id=151:modelizacion-de-la-distribucion-de-isotopos-radiactivos-y-farmacos-en-el-organismo&catid=13:seminarios&Itemid=225

MODELIZACIÓN DE LA DISTRIBUCIÓN DE ISÓTOPOS RADIACTIVOS Y FÁRMACOS EN EL ORGANISMO

Actualizado: 2016-06-07

Guillermo Sánchez. (http://diarium.usal.es/guillermo)

Notas:

La presentación está elaborada con el programa BIOKMOD. En la ayuda del programa están incluido la mayoria de los ejemplos.

Parte del material de este notebook procede del libro *Mathematica* más allá de las matemáticas, 2a Edición.disponible en GoogleStore: https://play.google.com/store/books/details?id=KjfeBQAAQBAJ

¿Dónde queremos llegar?

¿Cómo se modeliza un proceso biocinético/farmacocinético? : Modelos lineales y no lineales

¿Cómo se pueden obtener los parámetros del modelo experimentalmente?

¿Qué es el diseño óptimo?

¿Que son los modelos biocinetico de la ICRP y para que se utilizan?

¿Como se puede inferir la dosis a partir de bioensayos (Excreción urinaria, CRC, etc.)?

¿Como lo haremos?

Ayudandonos del programa BIOKMOD (requiere *Mathematica* 10) de donde procede el material utilizado y esta disponible el programa para descarga(http://diarium.usal.es/guillermo/biokmod/).

Algunos de los ejemplos pueder reproducirse directamente en la web con BiokmodWeb (http://www3.enusa.es/webMathematica/Public/biokmod.html)

Modelización compartimental (MC)

Sistema físico o biológico que se descompone en un número finito de componentes llamados compartimentos que intercambian materia (partículas o flujo) entre ellos y/o con el exterior



Algunos usos

Modelización del metabolismo de la incorporación de partículas inhaladas, ingeridas injectadas.

Modelización de la incorporación por ingestión o inyección de compuestos a personas y otros seres vivos en Medicina y en Farmacia.

La MC se aplica en otras áreas como en el Transporte de partículas en estudios medioambientales

Ejemplo sencillos

La desintegración radiactiva: A -> B



$$\frac{dx(t)}{dt} = -k\,x(t)$$

Organo aislado

Supongamos un recipiente de volumen V lleno de agua con sal con un concentración q=o en t = o. Hay una entrada agua salada, con caudal v y concentración c, constantes. Se va produciendo instantáneamente una mezcla de concentración q(t) que va saliendo también a un caudal v.



$$q + dq = \frac{\overbrace{V\,q}^{\text{Sal que habia}} + \overbrace{c\,v\,dt}^{\text{Sal que entra}} - \overbrace{q\,v\,dt}^{\text{sal que sale}}}{V} \Rightarrow \frac{dq}{dt} = \frac{v(c-q)}{V}$$

Modelo bicompartimental simple

Consideremos el modelo de la figura



Formulación matematica

Llamamos $x_1(t)$ y $x_2(t)$, $t \geq 0$, a la cantidad de la especie retenida en los compartimentos 1 y 2 , respectivamente.

$\frac{dx_1}{dt} =$ $b(t)$ $-$ drug distribution rate from 1 to 2

$\frac{dx_2}{dt} =$ inflow rate (from 1) $-$ outflow rate (elimination)

Esta ecuancion se conoce como **mass balance equation**. Si asumimos que la tasa de trasferencia, k_{12}, k_{20}, es proporcional a la masa (o concentracion) existente en el compartimento en t, entonces:

$$\frac{dx_1}{dt} = b(t) - k_{12}\, x_1$$

$$\frac{dx_2}{dt} = k_{12}\, x_1 - k_{20}\, x_2$$

En notacion matricial:

$$\begin{pmatrix} x_1'(t) \\ x_2'(t) \end{pmatrix} = \begin{pmatrix} -k_{12} & 0 \\ k_{12} & k_{20} \end{pmatrix} + \begin{pmatrix} b(t) \\ 0 \end{pmatrix}$$

Modelo bicompartimental generalizado

Formulación matemática: Con entrada y salida al exterior desde los compartimentos 1 y 2



Llamamos $x_1(t)$ y $x_2(t)$ a las variables de estado (concentración, cantidad, etc) y su evolución en el tiempo, entonces el sistema podemos describirlo por el sistema de ecuaciones diferenciales siguientes

$$\frac{dx_1}{dt} = -\overbrace{(k_{12}+k_{10})}^{K_{12}} x_1 + k_{21}x_2 + b_1(t)$$

$$\frac{dx_2}{dt} = k_{12}x_1 - \overbrace{(k_{21}+k_{20})}^{K_{21}} x_2 + b_2(t)$$

Podemos reformularlo en notación matricial como sigue:

$$\mathbf{x}'(t) = \mathbf{A}\,\mathbf{x}(t) + \mathbf{b}(t)$$

$$\mathbf{x}'(t) = \begin{pmatrix} x_1'(t) \\ x_2'(t) \end{pmatrix} \quad \mathbf{A} = \begin{pmatrix} -K_{12} & k_{21} \\ k_{12} & -K_{21} \end{pmatrix} \quad \mathbf{x}(t) = \begin{pmatrix} x_1(t) \\ x_2(t) \end{pmatrix} \quad \mathbf{b}(t) = \begin{pmatrix} b_1(t) \\ b_2(t) \end{pmatrix}$$

Ejemplos: Resolución con BIOKMOD

Con Biokmod podemos generar directamente el SEDO y su solución (requiere Mathematica 10). Se requiere cargar el programa

```
Needs["Biokmod`SysModel`"]
```

```
SysModel, version 1.5.1   2013-11-12
```

```
X[t1_, k12_, k10_, k21_, k20_] :=
 SystemDSolve[CompartMatrix[2, {{1, 2, k12}, {2, 1, k21}, {1, 0, k10}, {2, 0, k20}}],
  {1, 1}, {0, 0}, t, t1, x]
```

```
Manipulate[Plot[{x1[t1], x2[t1]} /. X[t1, k12, k10, k21, k20], {t1, 0, 10}],
 {{k12, 0.2}, 0.1, 1}, {{k21, 0.3}, 0.1, 1},
 {{k10, 0.4}, 0.1, 1},  {{k20, 0.5}, 0.1, 1}]
```



Ejemplo: Plantea el modelo general bicompartimental con Biokmod considerando la constante de desintegración λ

Lo primero es construir la matriz compartimental

```
model2 = CompartMatrix[2, {{1, 2, k12}, {2, 1, k21}, {1, 0, k10}, {2, 0, k20}}, λ]
```

```
{{-k10 - k12 - λ, k21}, {k12, -k20 - k21 - λ}}
```

La siguiente función no calcula nada pero permite generar automaticamente el sistema de ecuaciones difrenciales

```
ShowODE[model2, {x1[0], x2[0]}, {b1[t], b2[t]}, t, x] // TableForm
```

```
x1'[t] == b1[t] + (-k10 - k12 - λ) x1[t] + k21 x2[t]
x2'[t] == b2[t] + k12 x1[t] + (-k20 - k21 - λ) x2[t]
x1[0] == x1[0]
x2[0] == x2[0]
```

Ejemplo: Resuelve el modelo anterior para: $k_{10} = 0.05$, $k_{20} = 0.3$, $k_{12} = 0.4$, $k_{21} = 0.3$ y λ=0. Supón una incorporación continua constante b1 = 1 y b2 = 1, con condiciones iniciales: $x_1[0] = 0$, y $x_2[0] = 0$.

```
modelo = model2 /. {k10 → 0.05, k20 → 0.03, k12 → 0.04, k21 → 0.02 , λ → 0};
entradas = {1, 1};
condicionesiniciales = {0, 0};
{x1[t1_], x2[t1_]} =
 {x1[t1], x2[t1]} /. SystemDSolve[modelo, condicionesiniciales, entradas, t, t1, x]
```

$$\left\{18.9189 - 4.77824\, e^{-0.104641\, t1} - 14.1407\, e^{-0.035359\, t1},\ 35.1351 + 3.49792\, e^{-0.104641\, t1} - 38.6331\, e^{-0.035359\, t1}\right\}$$

```
Plot[{x1[t], x2[t]}, {t, 0, 100}]
```



La retención en el compartimento 1 o 2 puede calcularse para cualquier instante t

```
{x1[10], x2[10]}
```

```
{7.31176, 9.23689}
```

Las variables definidas si no las vamos a utilizar mas adelante podemos borrarlas:

```
Clear[model2, modelo, entradas, condicionesiniciales, x1, x2];
```

Generalización a n compartimentos



$\dot{\mathbf{x}}(t) = \mathbf{A}\,\mathbf{x} + \mathbf{b}(t),\ t \geq 0$

$\mathbf{x}(0) = \mathbf{x}_0$

donde:

$x(t) = \{x_1(t),\ x_2(t),\ \ldots,\ x_n(t)\}^T$ siendo $x_i(t)$ la cantidad (masa, desintegraciones, concentración, etc) en el compartimento i en función de t.

A es una matriz $n \times n$ conocida como matriz compartimental

$$\text{(para n = 3) A} = \begin{pmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{pmatrix};$$

$a_{11} = -(k_{10} + k_{12} + k_{13} + \lambda)$; $a_{12} = k_{21}$; $a_{13} = k_{31}$;

$a_{21} = k_{12}$; $a_{22} = -(k_{20} + k_{21} + k_{23} + \lambda)$; $a_{23} = k_{32}$;

$a_{31} = k_{13}$; $a_{32} = k_{23}$; $a_{33} = -(k_{30} + k_{31} + k_{32} + \lambda)$;

$\mathbf{b}(t) = \{b_1(t),\ b_2(t),\ \ldots,\ b_n(t)\}^T$ donde $\{b_i(t)\}$ es la entrada hacia el compartimento i desde el exterior del sistema

$\mathbf{x}(0) = \{x_1(0),\ x_2(0),\ \ldots,\ x_n(0)\}^T$ son las condiciones iniciales, esto es $x_i(0)$, representa la cantidad en el compartmento i en $t = 0$.

Ejemplo: Modelo del iodo. Resolucion con BiokmodWeb



Iodine biokinetic model -see the figure - where compartment 1 is the blood, compartment 2 is the thyroid , compartment 3 is the rest of the body, and "0" mean "out of the system" . The rate transfer values (clearance), in days^-1, taken from ICRP 78 are k10 = 1.9404, k12= 0.8316, k23 = Log[2]/80, k30 = 0.01155 and k31= 0.0462, k30 = 0.01155 and k40 = 12. Them the compartmental matrix is defined:{{1, 2, 0.83}, {1, 4, 1.94}, {2, 3, Log[2]/80}, {3, 0, 0.01155}, {3, 1, 0.0462}, {4, 0, 12}} . Also it assumed the radioactive decay constant of I-131 that is Log[2]/8.02 (days^-1). It assumed a continuouos input "1.2 Exp[-0.2 t]" in compartment 1, then the input is written: {1.2 Exp[-0.2 t], 0, 0, 0}. Also it is supossed as initial contition {1, 0, 0, 0}.

http://www3.enusa.es/webMathematica/Public/biokmod1.jsp

Solución

Enter the compartmental matrix:

```
{{1, 2, 0.83}, {1, 4, 1.94},
{2, 3, Log[2]/80}, {3, 0, 0.01155}, {3, 1, 0.0462}, {4, 0, 12}}
```

Number of compartments: 4 Decay constant: Log[2]/8 Initial conditions at time t = 0: {1,0,0,0}

Input function in each compartment : {1.2 Exp[-0.2 t],0, 0, 0}
All values of this field must be {0,.., 0} if it is an impulsive single-input, because the inputs are the initial conditions.

Time t to evaluate the content in each compartment (i.e.: t or {5, 20, 30}) : t

Range of t to be plotted: From t-min 0 to t-max 10

Time t, in days (kij mush be in days), to compute the acumulated disintegrations in each compartment : 50*365

Evaluate

Solution

$$x_1(t) \to 0.548312\, e^{-2.8566\, t} + 0.461614\, e^{-0.2\, t} - 0.0197996\, e^{-0.146776\, t} + 0.00987282\, e^{-0.0929699\, t}$$
$$x_2(t) \to -0.164814\, e^{-2.8566\, t} - 3.65968\, e^{-0.2\, t} + 0.3193\, e^{-0.146776\, t} + 3.50519\, e^{-0.0929699\, t}$$
$$x_3(t) \to 0.00052651\, e^{-2.8566\, t} + 0.570232\, e^{-0.2\, t} - 1.16135\, e^{-0.146776\, t} + 0.59059\, e^{-0.0929699\, t}$$
$$x_4(t) \to -0.188965\, e^{-12.0866\, t} + 0.115246\, e^{-2.8566\, t} + 0.0753393\, e^{-0.2\, t} - 0.00321706\, e^{-0.146776\, t} + 0.00159695\, e^{-0.0929699\, t}$$

Plot



Us (disintegrations in each compartment during a time t)

$$\{213\,522.,\ 1.85948 \times 10^{6},\ 111\,578.,\ 34\,271.9\}$$

Ejemplo: Modelo interactivo del I-131 que muestre la retención en el compartimento 1 en función de los parámetros k12 y k23

In[9]:=
```
Needs["Biokmod`SysModel`"]
```
```
SysModel, version 1.5.1   2013-11-12
```

Supondremos el modelo de iodo que antes hemos utilizado, en este casl queremos dejar la solución en funcion de los parámetros k12 y k23

In[10]:=
```
iodine131matrix = CompartMatrix[3, {{1, 2, k12}, {1, 0, 1.9404},
    {2, 3, k23} , {3, 0, 0.01155} , {3, 1, 0.0462}}, Log[2] / 8.02] ;
```

In[11]:=
```
binput = {-27.13 e^(-24.08 t) + 27.13 e^(-2.86 t) - 0.020 e^(-0.147 t) + 0.0194 e^(-0.093 t), 0, 0};
```

In[12]:=
```
{x1, x2, x3} = {x1 , x2, x3} /. ParametricSystemNDSolve[
    iodine131matrix, {0, 0, 0}, binput, {t, 0, 100}, x, {k12, k23}];
```

The retention in blood (compart 1) is plotted as function of parameter k12 and k23.

In[13]:=
```
Manipulate[LogPlot[x1[k12, k23][t], {t, 1, 100}],
 {{k12, 0.83, "k13"}, 0.5, 1.2}, {{k23, 0.0086625, "k23"}, 0.001, 0.02}]
```

Out[13]=



Pharmakinetic Modeling

(http://demonstrations.wolfram.com/PharmacokineticModeling/)

Contributed by: **Nicholas R. Larson** (With additional contributions by: **John L. Falconer** and **Rachael L. Baumann**

The one- and two-compartment pharmacokinetic models give the concentration profile of a drug in the body for a given dosing scheme. The area under these curves corresponds to the total exposure to the drug. The therapeutic window for a drug is the range between the effective dose and the toxic dose. The dosing scheme for a drug is designed to attain the target concentrations within the therapeutic window. The therapeutic window in this Demonstration is preset, but it is different for each drug. Drug elimination reactions can have varying reaction orders and associated rate constants depending on the system. Drugs are often absorbed in the first compartment when administered orally; the drug must first be absorbed in the stomach before entering the blood stream.



Ejemplo: Coeficientes de transferencia variables



En el modelo de la figura los coef. de trasferencia o microconstante son: $k12 \to 2$, $k21 \to 1$, k10→ 0.5, $k_g \to 0.1$, $k_a \to 2$, $k_{bilis,gi}$ tiene una trasferencia periodica durante 2 horas $k_{bilis,gi}$= 1 seguida de 6 horas sin trasferencia y el ciclo se repite. Resuelve el modelo,

Solución

```
h[t_, a_, d_] := 1 - UnitStep[Mod[t, a + d] - d];
Plot[h[t, 8, 2], {t, 0, 110}, PlotPoints → 1000, PlotRange → All, Exclusions → None]
```



Las constantes de trasferencia son: $\{k12 \to 2, k21 \to 1, k10 \to 0.5, k_g \to 0.1, k_a \to 2, k_{bilis,gi} \to h[t, 8, 2]\}$

```
matrixhepatic = CompartMatrix[4,
    {{1, 2, k12}, {1, 0, k10}, {2, 1, k21}, {1, 3, k_g}, {3, 4, k_bilis,gi}, {4, 1, k_a}}] /.
  {k12 → 2, k21 → 1, k10 → 0.5, k_g → 0.1, k_a → 2, k_bilis,gi → h[t, 8, 2]};
```

Condiciones iniciales: suponemos un imput x1(0) = 200 en el compartimento 1 en t = 0, y x2=x3=x4=0 en t= 0.

```
ic2 = {200, 0, 0, 0};
ip = {0, 0, 0, 0};
```

En este caso la solución será numerica por ello tenemos que usar: SystemNDSolve

```
sol2 = SystemNDSolve[matrixhepatic, ic2, ip, {t, 0, 100}, t, x, MaxSteps → 5000];
```

```
{x1[t_], x2[t_], x3[t_], x4[t_]} = {x1[t], x2[t], x3[t], x4[t]} /. sol2
```

```
{InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t],
 InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t],
 InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t],
 InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t]}
```

```
Plot[{x1[t], x2[t], x3[t]}, {t, 0, 50},
 PlotPoints → 1000, PlotRange → {0, 100}, AxesLabel → {"days", "r"},
 PlotLegends → Placed[{"x1", "x2", "x3"}, Center, (Framed[#, RoundingRadius → 5] &)]]
```



Debajo se muestran los valores numericos de { x1[t],x2[t],x3[t],x4[t]}

```
Table[{t, x1[t], x2[t], x3[t], x4[t]}, {t, 1, 100, 5}] // TableForm
```

1	48.572	99.9343	4.99672	1.93616
6	18.7684	45.5072	15.3896	0.000765492
11	11.9279	22.0627	8.33475	5.04145
16	5.38796	13.0615	7.00667	0.00095937
21	3.92517	6.71481	3.36405	2.05349
26	1.79818	4.35894	2.55851	0.000382508
31	1.35258	2.27341	1.20531	0.736876
36	0.621503	1.50656	0.900393	0.000136765
41	0.470606	0.788125	0.42263	0.258455
46	0.216373	0.524499	0.314604	0.0000479382
51	0.164059	0.274548	0.147563	0.0902459
56	0.0754393	0.182869	0.109768	0.0000167362
61	0.0572152	0.0957343	0.0514783	0.0314833
66	0.02631	0.0637768	0.0382879	5.83814×10^{-6}
71	0.0199553	0.0333888	0.0179555	0.0109813
76	0.00917634	0.0222439	0.0133544	2.03565×10^{-6}
81	0.00696006	0.0116454	0.00626264	0.00383014
86	0.00320054	0.00775827	0.00465784	7.11903×10^{-7}
91	0.00242756	0.00406169	0.00218433	0.0013359
96	0.0011163	0.00270597	0.00162457	2.48676×10^{-7}

Michaelis-Menten

One compartmental with impulsive input

The one compartmental Michaelis-Menten model is given by equation

$\dot{x}_1(t) = -\frac{V_m\, x_1[t]}{k_m + x_1[t]} + b_1\, u_1[t]$

$b_1\, u_1[t] = D_1\, \delta[t]$, this mean that the initial condition is $x_1[0] = D_1$. Here is the solution when $\{D_1 \to 1,\ k_m \to 1.2,\ V_m \to 15\}$

```
m[t_] = x1[t] /. DSolve[{x1'[t] + 15 x1[t] / (1.2 + x1[t]) == 0, x1[0] == 1}, x1[t], t] // Quiet
```

$$\left\{1.2\, W\left(0.833333\, \sqrt[6]{2.71828^{5.-75.\,t}}\right)\right\}$$

```
Plot[m[t], {t, 0, 1}, PlotRange → All]
```




```
Clear[x1, x2]
```

The bicompartmental Michaeles-Menten model given by equation

$\dot{x}_1(t) = -k_{21}\, x_1(t) + k_{12}\, x_2(t) - \frac{k_1\, x_1(t)}{k_2 + x_1(t)}$

$\dot{x}_2(t) = k_{21}\, x_1(t) - k_{12}\, x_2(t)$

$x_1(o) = b_o\ ,\ x_2(o) = o$

with clearance coeffs:

```
k1 = 0.2; k2 = 0.3; k12 = 0.03; k21 = 0.02;
eq1 = x1'[t] + k21 x1[t] - k12 x2[t] + k1 x1[t] / (k2 + x1[t]) == 0
```

$$x1'(t) + \frac{0.2\, x1(t)}{x1(t) + 0.3} + 0.02\, x1(t) - 0.03\, x2(t) = 0$$

```
eq2 = x2'[t] - k21 x1[t] + k12 x2[t] == 0
```

$$-0.02\, x1(t) + x2'(t) + 0.03\, x2(t) = 0$$

It can be solved using NDSolve

```
s = NDSolve[{eq1, eq2, x1[0] == 1, x2[0] == 0}, {x1[t], x2[t]}, {t, 0, 10}]
```

$$\left\{\left\{x1(t) \to \text{InterpolatingFunction}\left[\boxed{\text{Domain: } \{0.,\ 10.\} \quad \text{Output: scalar}}\right](t),\ x2(t) \to \text{InterpolatingFunction}\left[\boxed{\text{Domain: } \{0.,\ 10.\} \quad \text{Output: scalar}}\right](t)\right\}\right\}$$

```
Plot[Evaluate[{x1[t], x2[t]} /. s], {t, 0, 10},
 PlotLegends → Placed[{"x1", "x2"}, Center, (Framed[#, RoundingRadius → 5] &)] ]
```




```
Quit[]
```

Ejemplo: Modelos no lineales

```
Quit[]
```

Ejemplo 1

Here is solved 2 D Fick' s law of diffusion from the boundaries of a circle

```
Ω = ImplicitRegion[(x^2 + y^2 ≤ 10), {{x, -5, 5}, {y, -5, 5}}];

eq1 = D[u[x, y, t], t] == 0.0000072 * (D[u[x, y, t], x, x] + D[u[x, y, t], y, y]) - 1.2;

sol = NDSolve[{eq1, DirichletCondition[u[x, y, t] == 100, x^2 + y^2 == 10],
    u[x, 0, t] == 10, u[0, y, t] == 10, u[x, y, 0] == 10}, u, {t, 0, 10}, {x, y} ∈ Ω];

Animate[ContourPlot[u[x, y, t] /. sol,
  {x, y} ∈ Ω, PlotRange → {0, 10}, ClippingStyle → Automatic,
  ColorFunction → "DarkRainbow", PlotLegends → Automatic], {t, 0, 100}]
```

Ejemplo 2

Roberts-Rowland equation (J. Pharmacokim.Biopharm 14: 227-260 (1986))

Here is solved eq. (1) by Roberts-Rowland J. Pharmacokim.Biopharm 14: 227-260 (1986)

$$\frac{\partial^2 c}{\partial z^2} - \frac{1}{d_n}\frac{\partial c}{\partial z} - \frac{r_n}{d_n} c = \frac{1}{d_n}\frac{\partial c}{\partial z}, \text{ where } c = c(z, t)$$

In *Mathematica* notation can be wrote

```
D[c[z, t], t] == d_n D[c[z, t], z, z] - D[c[z, t], z] - r_n d_n c[z, t]
```

Note: Roberts-Rowland paper use capital letters. (i.e: C in place of c, R_n in place of r_n, etc.)

Here is solved the case with "mixed" boundary conditions

Initial condition (IC): t =0, c= 0, 0≤z≤1

Boundary conditions (BC): (The "mixed" boundary conditions): For z=0: c =δ(z), for z=∞: c=0, in other notation: $c(0) = \delta(t)$ y $c(\infty) = 0$.

IC: c[z,0]→0

Here is solved eq.(18) using Laplace Transform. (The right side terms of eq. 18 are denoted by eq1 and the left by eq2)

```
eq1 = LaplaceTransform[D[c[z, t], t], t, s];
eq2 = LaplaceTransform[d_n D[c[z, t], z, z] - D[c[z, t], z] - r_n d_n c[z, t], t, s];
```

Now we calculate their Laplace Transforms. The are denoted in capital letters (i.e: for c[z,0] the LT is C[z,s]). BC and IC are applied.

```
eq1LT = eq1 /. {LaplaceTransform[c[z, t], t, s] → C[z, s]}
-c[z, 0] + s C[z, s]
```

```
eq2LT = eq2 /. {LaplaceTransform[c^(1,0)[z, t], t, s] → C^(1,0)[z, s],
    LaplaceTransform[c^(2,0)[z, t], t, s] → C^(2,0)[z, s],
    LaplaceTransform[c[z, t], t, s] → C[z, s]}
```

$$-C[z, s]\, d_n\, r_n - C^{(1,0)}[z, s] + d_n\, C^{(2,0)}[z, s]$$

```
A26 = eq1LT == eq2LT /. {c[z, 0] → 0}
```

$$s\, C[z, s] == -C[z, s]\, d_n\, r_n - C^{(1,0)}[z, s] + d_n\, C^{(2,0)}[z, s]$$

```
LaplaceTransform[DiracDelta[z], z, s]
```

$$1$$

```
eq26b = DSolve[{s C[z] == -C[z] d_n r_n - C'[z] + d_n C''[z], C[0] == 1}, C[z], z]
```

$$\left\{\left\{C[z] \to e^{\frac{1}{2} z \left(\frac{1}{d_n}+\frac{\sqrt{1+4\, s\, d_n+4\, d_n^2\, r_n}}{d_n}\right)} + e^{\frac{1}{2} z \left(\frac{1}{d_n}-\frac{\sqrt{1+4\, s\, d_n+4\, d_n^2\, r_n}}{d_n}\right)} C[1] - e^{\frac{1}{2} z \left(\frac{1}{d_n}+\frac{\sqrt{1+4\, s\, d_n+4\, d_n^2\, r_n}}{d_n}\right)} C[1]\right\}\right\}$$

The integration constant C[1] can be calculated using the BC: $C_b(\infty) = 0$

```
Solve[eq26b[[1]][[1, 2]] == 0, C[1]];
c = C[1] /. Solve[eq26b[[1]][[1, 2]] == 0, C[1]][[1]] // Simplify
```

$$\frac{e^{\frac{z \sqrt{1+4\, s\, d_n+4\, d_n^2\, r_n}}{d_n}}}{-1 + e^{\frac{z \sqrt{1+4\, s\, d_n+4\, d_n^2\, r_n}}{d_n}}}$$

```
eq26b /. {C[1] → 0} // FullSimplify
```

$$\left\{\left\{C[z] \to e^{\frac{z \left(1+\sqrt{1+4\, d_n\, (s+d_n\, r_n)}\right)}{2\, d_n}}\right\}\right\}$$

```
Cout[s_] = C[z] /. %[[1]]
```

$$e^{\frac{z \left(1+\sqrt{1+4\, d_n\, (s+d_n\, r_n)}\right)}{2\, d_n}}$$

Now it can be evaluated $C_{out}(z, t)$ taken the inverse laplace transform

```
cOut[z_, t_] = -InverseLaplaceTransform[Cout[s], s, t] // FullSimplify
```

$$\text{ConditionalExpression}\left[\frac{e^{-\frac{(t-z)^2}{4\, t\, d_n}-t\, d_n\, r_n}\, z}{2\sqrt{\pi}\, \sqrt{\frac{t^3}{d_n^3}}\, d_n^2},\ z\, d_n < 0\right]$$

```
csol[z_, t_] = e^(-(t-z)^2/(4 t d_n) - t d_n r_n) z / (2 Sqrt[π] Sqrt[t^3/d_n^3] d_n^2);
```

This solution has a little difference with eq.(49) Roberts-Rowland J. Pharmacokim.Biopharm 14: 227-260 (1986): r_n was deleted (maybe an mistake in the printed transcription). This mistake has not consequences when r_n=0. as happen in Fig 3.(B) of Roberts-Rowland paper

Here is shown this Fig 3.(B) obtain with our method

$\{d_n \to \{0.01, 0.1, 0.2, 1, 2\}, r_n \to 0$ in z =1

```
Plot[Evaluate[csol[1, t] /. {dn → {0.01, 0.1, 0.2, 1, 2}, rn → 0}],
 {t, 0.01, 4}, PlotRange → {0, 4}]
```



Here is the 3-D representation as function of z and t

```
Plot3D[Evaluate[csol[z, t] /. {dn → 0.01, rn → 0}],
 {z, 1, 2}, {t, 0.1, 4}, PlotRange → {0, 4}]
```



```
Plot3D[Evaluate[csol[z, t] /. {dn → 0.1, rn → 0}],
 {z, 0.1, 2}, {t, 0.1, 4}, PlotRange → {0, 4}]
```



```
Clear[c]
```

Here is shown that the solution obtained with *Mathematica* is right

$$c[z\_, t\_] = \frac{e^{-\frac{(t-z)^2}{4\,t\,d_n}-t\,d_n\,r_n}\,z}{2\sqrt{\pi}\,\sqrt{\frac{t^3}{d_n^3}}\,d_n^2};$$

```
D[c[z, t], t] == dn D[c[z, t], z, z] - D[c[z, t], z] - rn dn c[z, t] // Simplify
```

```
True
```

Regresión. Ajuste de modelos 1

```
Clear["Global`*"]
```

(Solo disponible en Biokmod)

We have the model of the below figure. The rate transfer from compartment 2 to environment is known and its value is k_{20}= 0.05 d^{-1}. The transfer coefficients k_{12} and k_{21} are unknown. The experiment consists of an instantaneous injection b_1= 1 in compartment 1 and b_2= 0.5 in compartment 2 at t = 0 of the model. I



It supposed that in t = 0 the amount of substance in all compartment is "0". The amount in compartment 2 in different times (in days) was measured. Here are the data {t, x} (These data are been obtained by simulation)

```
list1 = {{0, 0.5}, {10, 0.30}, {20, 0.26}, {30, 0.24}, {40, 0.21},
    {50, 0.19}, {60, 0.17}, {70, 0.15}, {80, 0.135}, {90, 0.12}, {100, 0.11}};
```

We intend to estimate the parameters k_{12} and k_{21} by adjusting our model to the experimental data given in list1. We will proceed as follows

Step 1: The compartmental matrix of the model is defined as a function of parameters to be fitted

```
modelTwoCompart = CompartMatrix[2, {{1, 2, k12}, {2, 1, k21}, {2, 0, 0.05}}]
```

```
{{-k12, 0. + k21}, {0. + k12, -0.05 - k21}}
```

Step 2: The model is built by choosing the function of retention in compartment 2. In this case, how is a single-impulsive input, AcuteInput is used.

```
x2[t_, k12_, k21_] = x2[t] /. AcuteInput[modelTwoCompart, {1, 0.5}, t, x];
```

Step 3: Now it is fitted the coefficients k_{12} and k_{21} with the experimental data. Because `x2[t,k12,k21]` is an analytic expression NonlinearRegress can used.

We will need the *Mathematica* package NonlinearFit.

```
nlm = NonlinearModelFit[list1, x2[t, k12, k21], {{k12, 0.01, 0.5}, {k21, 1, 5}}, {t}]
```

```
FittedModel[ -0.230443 (0. + 1. e^(-0.363082 t)) + 0.230443 (0. + 1. e^(-0.0111678 t)) + 0.5 (0.801291 (0. + 1. e^(-0.363082 t)) + 0.198709 (0. + 1. e^(-0.0111678 t))) ]
```

To get the functional form of the `FittedModel` object, use `Normal`:

```
Normal[nlm] // ExpandAll // Chop
```

```
0.170202 e^(-0.363082 t) + 0.329798 e^(-0.0111678 t)
```

The result is returned as a `FittedModel` object, of which properties can be returned:

```
nlm[{"ParameterTable", "ANOVATable"}]
```

	Estimate	Standard Error	t-Statistic	P-Value
k12	0.0810964	0.0122265	6.63285	0.0000955992
k21	0.243154	0.0422701	5.75238	0.000275471

	DF	SS	MS
Model	2	0.641483	0.320741
Error	9	0.0000420101	4.66779×10^{-6}
Uncorrected Total	11	0.641525	
Corrected Total	10	0.124414	

Here the fitted function and the experimental data are shown:

```
Plot[x2[t, 0.081, 0.2431], {t, 0, 100},
 Epilog → {Hue[0], PointSize[0.02], Map[Point, list1]}]
```




```
Clear["Global`*"]
```

Se pueden consultar mas ejemplos, incluido modelos multirespuesta en la ayuda de BIOKMOD

Regresión. Ajuste de modelos 2

```
Quit[]
Needs["Biokmod`SysModel`"]
SysModel, version 1.5.1   2013-11-12
```



A. Sánchez-Navarro, C. Casquero, and M. Weiss, 'Distribution of Ciprofloxacin and Ofloxacin in the Isolated Hindlimb of the Rat', Pharmaceutical Research, 16: 587-591 (1999).

Below equation represents a physiological model where the drug is transported from V_p (vascular volume), by perfusate flow Q, to V_{Tu} (tissue water space) across a permeability barrier (permeability-surface product PS) and its binding is described by binding/unbinding constants k_{on}/k_{off}:

$$\dot{x}_1(t) = -\frac{Q+PS}{V_p} x_1(t) + \frac{PS}{V_p} x_2(t) + \frac{Q}{V_p} b_1(t)$$

$$\dot{x}_2(t) = \frac{PS}{V_{\text{Tu}}} x_1(t) - \left(\frac{PS}{V_{\text{Tu}}} + k_{\text{on}}\right) x_2(t) + k_{\text{off}} \frac{V_{\text{Tb}}}{V_{\text{Tu}}} x_3(t)$$

$$\dot{x}_3(t) = k_{\text{on}} \frac{V_{\text{Tu}}}{V_{\text{Tb}}} x_2(t) - k_{\text{off}} x_3(t)$$

Also, the input function is $\left\{\frac{Q}{V_p} b_1(t), 0, 0\right\}$ and the initial condition is {0, 0, 0},

The matrix A for this model, which will be used later, is

```
physiomodel = CoefMatrix[3, {{1, 1, -(Q/Vp + PS/Vp)}, {1, 2, PS/Vp}, {2, 1, PS/VTu},
    {2, 2, -(PS/VTu + kon)}, {2, 3, koff VTb/VTu}, {3, 2, kon VTu/VTb}, {3, 3, -koff}}] ;
```

```
ShowODE[physiomodel, {0, 0, 0}, {Q/Vp b1[t], 0, 0}, t, x] // TableForm // TraditionalForm
```

$$x_1'(t) = \frac{Q\, b_1(t)}{V_p} + x_1(t)\left(-\frac{PS}{V_p} - \frac{Q}{V_p}\right) + \frac{PS\, x_2(t)}{V_p}$$
$$x_2'(t) = \frac{k_{off}\, x_3(t)\, V_{Tb}}{V_{Tu}} + x_2(t)\left(-k_{on} - \frac{PS}{V_{Tu}}\right) + \frac{PS\, x_1(t)}{V_{Tu}}$$
$$x_3'(t) = \frac{k_{on}\, x_2(t)\, V_{Tu}}{V_{Tb}} - k_{off}\, x_3(t)$$
$$x_1(0) = 0$$
$$x_2(0) = 0$$
$$x_3(0) = 0$$

The mathematical expression of $b_1(t)$ was unknown but an experiment was made where $b_1(t)$ was given by the best fit of the input function to experimental data $\{\{t_1, b_1\}, \ldots, \{t_n, b_n\}\}$ obtained via sampling from an arterial catheter. Here are the experimental data:

```
dataCateter = {{0., 0.}, {0.05, 402.7}, {0.1, 430.3}, {0.15, 375.4},
   {0.2, 292.4}, {0.25, 202.2}, {0.3, 148.4}, {0.35, 96.4}, {0.4, 64.9},
   {0.45, 41.7}, {0.5, 25.3}, {0.55, 17.8}, {0.6, 8.8}, {0.65, 6.6}, {0.7, 3.2},
   {0.75, 2.5}, {0.8, 1.4}, {0.85, 0.9}, {0.9, 0.5}, {1., 0.2}, {1.1, 0.07},
   {1.2, 0.03}, {1.3, 0.01}, {1.4, 0.003}, {1.45, 0.001}, {1.5, 0.001}};
ListPlot[dataCateter, Joined → True, PlotRange → All]
```



The graphic form suggests that the below function could be appropriated to fit the data

```
b[t_] = a t Exp[c t] /. FindFit[dataCateter, a t Exp[c t] , {a, c}, t]
```

```
13 610.1 e^-11.216 t t
```

Here are shown experimental data and the fitted function

```
Plot[b[t], {t, 0, 1}, Epilog → {Hue[0], PointSize[0.02`], Point /@ dataCateter}]
```



Now, we used the kown values for $V_{Tu} \to 6.411,\ V_p \to 0.973,\ PS \to 2.714,\ Q \to 3,\ V_{Tb} \to 1.$

```
inputb[t_] = {Q/Vp b[t] , 0, 0} /. {Vp → 0.973, Q → 3}
```

```
{3.08325 b[t], 0, 0}
```

```
physiomodel1[kon_, koff_] =
  CoefMatrix[3, {{1, 1, -(Q/Vp + PS/Vp)}, {1, 2, PS/Vp}, {2, 1, PS/VTu}, {2, 2, -(PS/VTu + kon)}},
```

$$\left\{2,\ 3,\ k_{off}\ \frac{V_{Tb}}{V_{Tu}}\right\},\ \left\{3,\ 2,\ k_{on}\ \frac{V_{Tu}}{V_{Tb}}\right\},\ \{3,\ 3,\ -k_{off}\}\Big]\ /.$$

```
{V_Tu → 6.411, V_p → 0.973, PS → 2.714, Q → 3, V_Tb → 1, k_on → kon, k_off → koff };
```

It was also measured $x_1(t)$ by sampling and it was obtained the following experimental data $\{\{t_1, x_1(t_1)\}, \ldots, \{t_n, x_1(t_n)\}\}$.

```
dataPhysio = {{0.03, 14.50}, {0.08, 61.06}, {0.28, 120.93}, {0.33, 109.01},
   {0.38, 98.08}, {0.48, 69.66}, {0.55, 51.51}, {0.65, 34.77}, {0.75, 23.21},
   {0.85, 15.59}, {0.95, 12.16}, {1.05, 9.598}, {1.15, 8.278}, {1.25, 6.842},
   {1.35, 5.871}, {1.45, 5.297}, {1.6, 4.886}, {1.8, 3.846}, {2., 3.317},
   {2.2, 2.899}, {2.4, 2.627}, {2.6, 2.289}, {2.8, 1.998}, {3., 1.930},
   {3.4, 1.589}, {3.75, 1.308}, {4.25, 1.112}, {4.75, 1.064}, {5.25, 0.938},
   {6.75, 0.842}, {7.25, 0.831}, {7.75, 0.778}, {8.25, 0.818}, {11., 0.739}};

model[t1_?NumericQ, kon_?NumericQ, koff_?NumericQ] := x1[t1] /.
  SystemNDSolve[physiomodel1[kon, koff], {0, 0, 0}, inputb[t] , {t, 0, 12}, t1, x]
```

Finally we can apply NonlinearModelFit including the statistical reports (the list of available reports can be check `nlm["Properties"]`). It will usually take a long time.

```
nlm = NonlinearModelFit[dataPhysio , model[t, kon, koff],
    {{kon, 0.71}, {koff, 0.11}}, {t}, Method → Gradient];

Show[ListPlot[dataPhysio],
 Plot[nlm[t], {t, 0, 12}, PlotRange -> {0, 500}], Frame → True]
```



```
nlm["ParameterTable"]
```

	Estimate	Standard Error	t-Statistic	P-Value
kon	0.71354	0.135844	5.25264	9.52803×10^{-6}
koff	0.110548	0.0870832	1.26945	0.213436

```
nlm["FitResiduals"]
```

```
{0.25997, -2.44327, -6.16585, -5.08627, -0.483875, 1.01401, -0.291513, 0.404163,
 -0.00898464, -0.751601, 0.0269447, 0.0759773, 0.430413, 0.122153, -0.0434651,
 -0.00695015, 0.281614, -0.0567019, -0.0409613, -0.0203854, 0.0657251,
 0.0217401, -0.0272953, 0.104098, 0.063412, -0.0339379, -0.0528448, 0.0114619,
 -0.0418627, -0.0314969, -0.0237137, -0.0614143, -0.00828895, -0.0301865}
```

```
ListPlot[%, Filling → Axis]
```



```
ClearAll[physiomodel, physiomodel1, dataCateter, b, inputb, X1, x1fit, model];
```

Diseño óptimo

```
Needs["Biokmod`Optdesign`"]
```

```
Optdesign, 1.0   2007-04-09
```

```
Optdes[ inp (e^(-2.0 t-0.09 p) + e^(-0.001 t-0.2 p)), t, {{inp, 100}, {p, 5}} , 0.5, 1, 2, 1]
```

```
{0.055815, {t0 → 0.5, t1 → 3.95733}}
```

Modelos ICRP



Fig. 3.1. Illustration of the movement of radionuclides in the human body.

Modelo ICRP 66 interactivo

```
Clear["Global`*"]
```

o pueden resolverse que Biokmod (esta en la ayuda eligiendo ICRP 66)

```
In[3]:= Needs["Biokmod`Resptract`"] // Quiet
Resptract 1.2  2005-05-16

In[4]:= Manipulate[met = ToExpression[metabolization];
 LogPlot[LungsRetention[int, AMADfit[amad] , met, t, lambda],
  {t, 0, tmax}, AxesLabel → {"days", "kBq"}, PlotRange → All],
{{tmax, 10, "days"}, 1, 100, Appearance → "Open"}, {{int, 10, "kBq"}, 1, 100,
 Appearance → "Open"}, {{amad, 5, "AMAD(μm)"}, 1, 10, Appearance → "Open"},
{metabolization, {"S", "M", "F"}, ControlType → PopupMenu},
{lambda, 0, ControlType → InputField}, SynchronousUpdating → False]
```



(La constante de desintegración para cada isótopo puede obtenerse como se ha visto antes, si es muy pequeña puede tomarse “o” como ocurre con los isótopos U238, U235, U234)

Debajo se comparan la retención pulmonar para las diferentes solubilidades (la retención pulmonar para tipo F es casi nula):

```
{lung5S[t_], lung5M[t_], lung5F[t_]} =
 LungsRetention[1, AMADfit[5] , #, t, 0] & /@ {S, M, F} // Chop;
Plot[{lung5S[t], lung5M[t], lung5F[t]}, {t, 1, 3},
 PlotStyle → {Orange, Dashed, Thick}, PlotLegends → Placed[{"S", "M", "F"}, Right]]
```



Pueden resolverse con BiokmodWeb:

http://www3.enusa.es/webMathematica/Public/biokmod.html ->ICRP Models ->Lung

Modelo ICRP 78.- Bioensayos

Uno de las aplicaciones de la modelización compartimental es la modelización de bioensayos (por ej: excreción urinaria, fecal o retención pulmonar o en todo el cuerpo). Se trata de estimar la cantidad incorporada a partir del bioensayo U(En la grafica se representa el modelo GI de la ICRP 30, sustituido por la ICRP Publication 100, en la practica se sigue utilizando la ICRP30).



B y RB representa conceptualmente el modelo especifico de cada elemento.

Los modelos de la ICRP 78 están incluidos en:

http://www3.enusa.es/webMathematica/Public/biokmod.html ->ICRP Models ->General

o pueden resolverver que Biokmod

```
Needs["Biokmod`Biokdata`"]
```

```
Humorap 3.7 2015-06-15
```

```
iodine::shdw : Symbol iodine appears in multiple contexts {Biokmod`Biokdata`, Global`}; definitions in context Biokmod`Biokdata` may shadow or be shadowed by other definitions. >>
```

```
Biokdata 1.2.2 2004-09-3
```

Ejemplo: Modelo del iodo incluyendo el TGI (B es el compartimento 1)



The halt-life of the blood is 0.25 days, the uptake to thyroid is 30% and to the bladder is 70%, the halt-life of Thyroid is 80 days, and the halt-life of "Rest of Body" is 12 days to 20 % faecal excretion and 80% to blood, therefore the rate transfer are, in d^{-1}: Blood (Log[2]/0.25), compartment 1, Thyroid (Log[2]/80), compartment 2 and Rest of body (Log[2]/12, compartment 4. As usual, the rate transfers from bladder is 12 d^{-1}, also it mush be added URI (compartment 5), ULI(compartment 6), LLI (Compartment 7) and FEC (compartment 8) .

Con la inforamación anterior se construye la matriz de trasferencia:

```
TableForm[TransferRates[iodine],
 TableHeadings → {None, {"From", "to", "Tranfer Rates(days^-1)"}}]
```

From	to	Tranfer Rates(days^{-1})
1	2	0.831777
1	4	1.94081
2	3	$\frac{Log[2]}{80}$
3	6	0.0115525
3	1	0.0462098
4	5	12
6	7	k_{ULI}
7	8	k_{LLI}

```
iodineMatrix = CompartMatrix[8, {{1, 2, 0.83}, {1, 4, 1.94}, {2, 3, Log[2]/80},
    {3, 6, 0.01155}, {3, 1, 0.0462}, {4, 5, 12}, {6, 7, kULI}, {7, 8, kLLI}}];
```

En el caso de una incorporación puntual de iodio 131 la función de retención es (para el iodo f1 = 1, y para I131 T1/2 = 8.02 d),

```
BioassayReport[iodineMatrix, 1, t1, Log[2] / 8.02,
  IntakeWay → "Injection", IntakeType → "Acute"] // TableForm
```

qDailyUrine[t1] → $-34208.3\, e^{-12.0864\, t1} + 13.6205\, e^{-2.85638\, t1} - 0.00168984\, e^{-0.146559\, t1} + 0.00157422$
qDailyFaecal[t1] → $0.0000627959\, e^{-2.85638\, t1} - 0.000173203\, e^{-1.88643\, t1} + 0.000194267\, e^{-1.08643\, t1} - 0$
qWholebody[t1] → $-0.210184\, e^{-12.0864\, t1} + 0.910589\, e^{-2.85638\, t1} - 0.0000343\, e^{-1.88643\, t1} + 0.000113059$

Todos los pasos anteriores podemos obvierlos y usar directamente (las pequeñas diferencias son debidas a que se emplean mas decimales en los coef. de trasferencia):

```
BiokdataReport[iodine, "Injection", "Acute", "Automatic", 1, 1, t1,
  Log[2] / 8.02, CompartNumbers → "True", TransferRates → "True"] // Chop
```

{{qDailyUrine[t1] →
$-34232.2\, e^{-12.0864\, t1} + 13.6549\, e^{-2.85897\, t1} - 0.00169077\, e^{-0.146574\, t1} + 0.00157512\, e^{-0.0927512\, t1}$,
qDailyFaecal[t1] → $0.0000626808\, e^{-2.85897\, t1} - 0.00017315\, e^{-1.88643\, t1} +$
$0.000194443\, e^{-1.08643\, t1} - 0.000647024\, e^{-0.146574\, t1} + 0.000566524\, e^{-0.0927512\, t1}$,
qWholebody[t1] → $-0.210331\, e^{-12.0864\, t1} + 0.910375\, e^{-2.85897\, t1} - 0.0000342894\, e^{-1.88643\, t1} +$
$0.000113161\, e^{-1.08643\, t1} - 0.0377108\, e^{-0.146574\, t1} + 0.337588\, e^{-0.0927512\, t1}$},

Compart numbers →

1	Blood
2	Thyroid
3	Rest
4	Bladder
5	Urine
6	ULI
7	LLI
8	FEC

,

Transfer Rates →

From	to	Tranfer Rates(days^{-1})
1	2	0.831777
1	4	1.94081
2	3	$\frac{Log[2]}{80}$
3	6	0.0115525
3	1	0.0462098
4	5	12
6	7	k_{ULI}
7	8	k_{LLI}

}}

Bioensayos. Estimación de la cantidad incorporada a partir de un bioensayo

Idealmente a partir de un bioensayo podemos estimar la cantidad incorporada: Supuesto el modelo estándar y el resultado de una muestra, m tomada ti tras una incorporación puntual (sin incertidumbre estadística) podemos determinar la cantidad incorporada

$$\mathbf{q(t) = I\, r(t),\ con\ q(ti) = m \rightarrow I = m/r(ti)}$$

Ejemplo: A un paciente se le ha penetrado a través de una herida (modelo inyección) una cantidad desconocida de iodo 131. 5 dias despues se mide 5.4 kBq en una muestra urinaria de 24 horas ¿Que cantidad incorporó?

```
BiokdataReport[iodine, "Injection",
  "Acute", "Automatic", 1, 1, t1, Log[2] / 8.02] // Chop
```

```
{qDailyUrine[t1] →
  -34 232.2 e^-12.0864 t1 + 13.6549 e^-2.85897 t1 - 0.00169077 e^-0.146574 t1 + 0.00157512 e^-0.0927512 t1,
 qDailyFaecal[t1] → 0.0000626808 e^-2.85897 t1 - 0.00017315 e^-1.88643 t1 +
  0.000194443 e^-1.08643 t1 - 0.000647024 e^-0.146574 t1 + 0.000566524 e^-0.0927512 t1,
 qWholebody[t1] → -0.210331 e^-12.0864 t1 + 0.910375 e^-2.85897 t1 - 0.0000342894 e^-1.88643 t1 +
  0.000113161 e^-1.08643 t1 - 0.0377108 e^-0.146574 t1 + 0.337588 e^-0.0927512 t1}
```

De la salida anterior copiamos lo correspondiente a la excrección urinaria y calculamos I

```
r[t_] = 34 208 e^-12.0866 t + 13.620 e^-2.85656 t - 0.00168983 e^-0.14677 t + 0.001574 e^-0.09297 t;
5.4 / r[5] "kBq"
```

```
29 008.3 kBq
```

Evaluación de bioensayos. Incertidumbres

The estimation of isotope content in a compartment or region involves many uncertainties even assuming that the ICRP metabolic models are a good representation of the real behaviour of the particles intake in the human body. This is so because most of the true values of the parameters at a real situation are unknown. The parameters usually applied are based on the reference values given in ICRPs.
Let's be $r(t)$ expressed as function of certain parameters $\{k_1, \ldots, k_r\}$ with their associated uncertainties: $\{u(k_1), \ldots u(k_r)\}$, then

$r(t) = F(k_1, \ldots, k_n, t) \pm u_C(t)$
being $u_C(t)$ the combined standard uncertainty.
Assuming that $\{k_1, \ldots, k_r\}$ are uncorrelated and taking the first-order Taylor series terms of $F(k_1, \ldots, k_n, \lambda, t)$, then $u_C(t)$ can be evaluated using

$$u_c^2(r(t)) = \sum_{i=1}^{r} \left(\frac{\partial F}{\partial k_i}\right)^2 u^2(k_i)$$

This is the expression used by BIOKMOD.

Lung retention predicted for a single intake of 1 Bq at t = 0, type S, decay constant negligible ($\lambda_R \to 0$) and AMAD p = 5 μm and $u_p = s_p$ = 0.5 μm . The dashed lines represent the confidence interval (95%) associated with the AMAD uncertainties.

```
rLung[p_, t_] = LungsRetention[1, AMADfit[p], S, t, 0] // Chop;
```

The evolution of the content with their associated uncertainties for a coverage factor k = 2 is computed and represented as follow

```
yu[t1_] =
  {"mean", "uL", "lL"} /. Uin[rLung[p, t1], {p} , {σp} , 2] /. {p → 5, σp → 0.5};
Plot[Evaluate[yu[t]], {t, 1, 100}, PlotRange → {0, 0.1}, AxesLabel → {"days", "Bq"}]
```



It can be observed that a small difference in the AMAD value has an important consequence in the lung retention predicted. For this reason, when the value for AMAD is

used to evaluate bioassay data and it it is not known then the intake estimated could have important uncertainties.

¿Qué es la dosis? ¿Qué implica la hipotésis lineal?

Las incorporaciones se convierten en dosis ¿pero que es la dosis? ¿Por qué las dosis son aditivas linealmente?.



Consecuencia de la hipotesis lineal (para un mismo isotopo y tipo de incorporación): Un incorporación única de 365 Bq produce la misma dosis que si se incopora 1 Bq/dia durante 365 dias.

Dosis interna. Definición de dosis comprometida efectiva

Concepto

The committed equivalent dose $H_T(\tau)$ is defined as the time integral of equivalent dose rate in a particular tissue or organ that will be receive by an individual following intake of radioactive material into the body, where τ is the integration time in years following the intake. Usually is taken τ = 50 y for adult .



$H_T(\tau)$ to organ or target tissue, T, as consequence of disintegration of a radionuclide in a source region, S, is given by (S and T can be the same compartment)

$$H_T(\tau) = \sum_R U_s(\tau)\, \mathrm{SEE}(T \leftarrow S)$$

where:

$U_s(\tau)$ = Número de transformaciones nucleares (Bq s) en τ años (por defecto τ = 50 para adultos) en el órgano o region S seguida a una incorporacion puntual.

$$U_s(\tau) = \int_0^\tau x(t)\, dt$$

x_j (t) es la actividad, en Bq, del nucleido R presente en el compartimento j en el instante t.

SEE(T←S) en Sv/(Bq s), es la energía específica efectiva que se deposita en T como consecuencia por cada desintegración (o trasformación nuclear), siguientes a una incorporación puntual en producida en los j órganos fuente S, donde representa los órganos o tejidos cuyas radiaciones llegan a T, incluida las que se producen en T. Se expresa en Sv/(Bq s).

$$\mathrm{SEE}(T \leftarrow S) = \frac{1.6 \times 10^{-13}}{m_T} \sum_R Y_R\, E_R\, w_R\, \mathrm{AF}(T \leftarrow S)_R$$

Y_R = yield of radiation R per nuclear transformation $(\mathrm{Bq}\ s)^{-1}$
E_R = energy of radiation R (MeV/Transformation)
$1.6 \times 10^{-13}\ \left(\frac{J}{\mathrm{MeV}}\right)$ is used for transform MeV in J
w_R = radiation weighing factor for radiation R (see table 8 of ICRP 68, typical values are: 20 for alpha particles and 1 for photons and electrons)
$\mathrm{AF}(T \leftarrow S)_R$= absorbed fraction in T per transformation in S for radiation R
m_T = mass of target tissue, T (kg).

When a nuclei decay in other nuclei which itself is radioactive, the contribution of the decay product to the dose is evaluated using a similar set of equations for the decay products. It usually is assumed that these decay products adopted the biokinectic of their parents.

Definición (The committed effective dose) (from ICRP 68 and ICRP 2001) :

The committed effective dose over a period τ (50 year for adult) is obtained as the sum of the weighted committed equivalent doses to organs or tissues.

$$e(\tau) = \sum_R H_T(\tau)\, W_T$$

Dosis interna. Calcula las desintegraciones acumuladas (Us) para I-131 en los distintos organos 1, 10 y 100 después de una inyección puntual y los factores de dosis equivalente efectiva.

```
Quit[]
```

Este ejemplo como en otras ocasiones pueden calcularse con BiokmodWeb o con Biokmod

http://www3.enusa.es/webMathematica/Public/biokmod.html ->Doses

```
Needs["Biokmod`Doses`"] // Quiet
```

```
 Doses 1.2   2015-08-27
```

Estudiemos la cadena de desintegracion del iodo 131 (se requiere conexión a internet).

```
IsotopeDecayModes["Iodine131"]
```

```
 BetaDecay
    β⁻
   1.00
 970.848 keV
```

```
IsotopeChainPlot["Iodine131"]
```


```
IsotopeChainData["Iodine131"]
```

	Half Life (s)	Daughter Nuclides	BranchingRatios
^{131}I	6.9338×10^{5} s	{ xenon-131 }	{1.00}
^{131}Xe	∞	{}	{}

It can be obtained the accumulated disintegration for different times: 1, 10 and 100 (in days after the acute intake by inhalation happened)

```
Disintegrations["I", "Injection", 1, 1, {1, 10, 100} ,
  Decayconst["I 131"], DisintegrationReport → "True"] // Chop
```

Compartment	day	10 day	100 day
ST	0		
SI	0		
B	28 486.5	30 293.3	30 475.7
ULI	0.122826	27.8161	97.6904
LLI	0.0423318	39.8548	161.818
Thyroid	16 228.1	158 721.	265 946.
UB_Content	4488.03	4864.2	4893.64
Other	54.1284	4898.09	15 954.1

```
CommittedDose["I 131", "Injection", 1, 1, 50 * 365.25]
```

Accumulted disintegration, in Bq, as function of the time

Compartment	18 262.5 day
ST	0.
SI	0.
B	30 475.8
ULI	97.7169
LLI	161.866
Thyroid	265 971.
UB_Content	4893.65
Other	15 958.2

Dose accumulated, in Sv, as function of the time

Sv/Bq	18 262.5 day
Testes	4.03538×10^{-11}
Ovarius	4.80886×10^{-11}
Red Marrow	1.00785×10^{-10}
Colon	5.69096×10^{-11}
Lungs	1.03046×10^{-10}
St Wall	4.28633×10^{-11}
Bladder Wall	7.65253×10^{-10}
Mama	5.86911×10^{-11}
Liver	4.75816×10^{-11}
Oesophagus	1.53732×10^{-10}
Thyroid	4.35955×10^{-7}
Skin	6.88474×10^{-11}
Bone Surface	1.32242×10^{-10}
Muscle	1.26693×10^{-10}
Brain	1.45286×10^{-10}
Small intestine	4.34103×10^{-11}
Kidneys	4.33654×10^{-11}
Pancreas	4.92598×10^{-11}
Spleen	4.63269×10^{-11}
Thymus	1.53732×10^{-10}
Uterus	5.74355×10^{-11}
Adrenals	4.80134×10^{-11}
Extrathoracic airways	1.48732×10^{-10}
Effective, e(50)	2.19033×10^{-8}

Material adicional:
http://diarium.usal.es/guillermo

http://diarium.usal.es/guillermo/biokmod/

Mathematica más allá de las matemáticas. 2ª Edición (marzo 2015, actualizado a Mathematica 10). Disponible en GoogleBooks y Playstore

Tutoriales y presentaciones en youtube: http://diarium.usal.es/guillermo/mathematica/

Bibliografia: http://diarium.usal.es/guillermo/publicaciones/especializadas/

Sobre Biokmod, bioensayos y modelización compartimental

Sánchez G; "Fitting bioassay data and performing uncertainty analysis with BIOKMOD" Health Physics.. 92(1) :64-72. 2007. ISSN/ISBN: 0017-9078

Sánchez G; Biokmod: A Mathematica toolbox for modeling Biokinetic Systems". Mathematica in Education and Research: 10 (2) 2005. ISSN/ISBN: 1096-3324

Lopez-Fidalgo J; Sánchez G; Statistical Criteria to Establish Bioassay Programs. Health Physics.. 89 (4). 2005. ISSN/ISBN: 0017-9078

Sánchez G; Lopez-Fidalgo J "Mathematical Techniques for Solving Analytically Large Compartmental Systems" Health Physics..: 85 (2): 2003. ISSN/ISBN: 0017-9078

Sobre diseño óptimo:

Juan M. Rodríguez-Díaz;Guillermo Sánchez-León:"Design optimality for models defined by a system of ordinary differential equations" Biometrical Journal 56 (5), pag 886–900, September 2014

G. Sánchez; J. M. Rodríguez-Díaz . Optimal design and mathematical model applied to establish bioassay programs": Radiation Protection Dosimetry. doi:10.1093/rpd/ncl499. 2007. ISSN/ISBN: ISSN 1742-3406

Lopez-Fidalgo J Rodri1guez-Díaz J.M.,Sánchez G; G., Santos-Martín M.T. Optimal designs for compartmental models with correlated observations" Journal of Applied Statistic. 32, 2006 ISSN/ISBN: 0266-4763